
Abstract We present a proof procedure for univariate real polynomial problems
in Isabelle/HOL. The core mathematics of our procedure is based on univariate
cylindrical algebraic decomposition. We follow the approach of untrusted certificates, separating solving from verifying: efficient external tools perform expensive
real algebraic computations, producing evidence that is formally checked within
Isabelle’s logic. This allows us to exploit highly-tuned computer algebra systems
like Mathematica to guide our procedure without impacting the correctness of its
results. We present experiments demonstrating the efficacy of this approach, in
many cases yielding orders of magnitude improvements over previous methods.

1 Introduction
Nonlinear polynomial systems are ubiquitous in science and engineering. As realworld applications of formal verification continue to grow and diversify, there is
an increasing need for proof assistants (e.g., ACL2, Coq, Isabelle [27], HOL Light
and PVS) to provide automation for reasoning about nonlinear systems over the
reals [25, 17, 24].

Cylindrical algebraic decomposition (CAD) [8] is one of the most powerful
known techniques for analysing non-linear polynomial systems. CAD-based methods have been implemented in various systems such as Z3 [9], QEPCAD [3], Mathematica and Maple. However, implementing CAD-based decision procedures within
proof assistants has been hindered by the difficulty in formalising the mathematics
justifying CAD computations.
In this paper, we present a formally verified procedure1 based on CAD for
univariate polynomial problems with rational coefficients. Goals such as
∀x. (x2 > 2 ∧ x10 − 2x5 + 1 ≥ 0) ∨ x < 2
∃x. (x2 = 2 ∧ (x > 1 ∨ x < 0))
can be discharged by our tactic automatically. It should be noted that certifying
a general multivariate CAD procedure is much harder, and the univariate version
we describe in the paper is only a first step in that direction.
A key feature of our procedure is its certificate-based design in which an external untrusted (but ideally highly efficient) program is used to find certificates,
and those certificates are then checked by verified internal procedures. Overall, the
soundness of our procedure depends solely on the soundness of Isabelle’s logic (and
code generation2 ) rather than trusted external oracles. This is much like Isabelle’s
sledgehammer tactic, which sceptically incorporates various external tools.
Our main contributions are:
– an efficient formalised theory of Tarski queries,
– an efficient approach to univariate sign determination at real algebraic points,
– a practical formally verified procedure for real algebraic problems based on
univariate CAD.
The paper continues at follows: A motivating example (§2) and a description of
the overall design (§3) sketch the general idea of our procedure. The construction
and manipulation of real algebraic numbers is developed in (§4), including a sign
determination procedure for evaluating polynomials at real algebraic points (§5).
The main proof is described in (§6), which is followed by a discussion of interaction
with external solvers (§7). Next, experiments and related work (§8) are described
along with further discussion of our tactic (§9). We then conclude with a look
towards the future (§10).

2 A Motivating Example
Unlike the general case of Rn , the restriction of CAD to univariate problems (i.e.,
to R1 ) is relatively straight-forward. Suppose we wish to prove
∀x. P (x) > 0 ∨ Q(x) ≥ 0
where
P (x) =
1

1 2
x −1
2

Code is available from https://bitbucket.org/liwenda1990/src_jar_2017
As our tactic is computationally intense, our procedure makes use of the proof by reflection
technique [16].

Fig. 1: The plot of P (x) = 21 x2 − 1 and Q(x) = x + 3

Q(x) = x + 3.

To do so, we can decompose R into disjoint connected components induced by
the roots of P and Q. This is illustrated in Fig. 1:
√
√ √
√
√
√
D = {(−∞, −3), { −3 }, (−3, − 2), { − 2 }, (− 2, 2), { 2 }, ( 2, ∞)}

root of Q

roots of P

and it can be observed that both P and Q have invariant signs over each of these
components. For
as can be seen from Fig. 1, P (x) < 0 and Q(x) > 0 hold
√ example,
√
for all x ∈ (− 2, 2). To decide the conjecture, we can pick sample points from
each of these components and evaluate λx. P (x) > 0 ∨ Q(x) ≥ 0 at these points.
That is,

∀x. P (x) > 0 ∨ Q(x) ≥ 0
= ∀D ∈ D. ∀x ∈ D. P (x) > 0 ∨ Q(x) ≥ 0
√
√
= ∀x ∈ {−4, −3, −2, − 2, 0, 2, 2}. P (x) > 0 ∨ Q(x) ≥ 0
= (P (−4) > 0 ∨ Q(−4) ≥ 0) ∧ (P (−3) > 0 ∨ Q(−3) ≥ 0) ∧ . . .
∧ (P (2) > 0 ∨ Q(2) ≥ 0)
= True

(1)

since
−4 ∈ (−∞, −3)
−3 ∈ {−3}

√
−2 ∈ (−3, − 2)
√
√
− 2 ∈ {− 2}
√ √
0 ∈ (− 2, 2)
√
√
2 ∈ { 2}
√
2 ∈ ( 2, ∞).
Analogously, to decide an existential formula
∃x. P (x) = 0 ∧ Q(x) > 0,
we have
∃x. P (x) = 0 ∧ Q(x) > 0
= ∃D ∈ D. ∃x ∈ D. P (x) = 0 ∧ Q(x) > 0
√
√
= ∃x ∈ {−4, −3, −2, − 2, 0, 2, 2}. P (x) = 0 ∧ Q(x) > 0
= (P (−4) = 0 ∧ Q(−4) > 0) ∨ (P (−3) = 0 ∧ Q(−3) > 0) ∨ . . .

(2)

∨ (P (2) = 0 ∧ Q(2) > 0)
= True.
In performing these arguments, there were a few “obvious” subtleties:
– The decomposition of R into the seven regions given covered the entire real
line. That is,
√
√
√ √
√
√
(−∞, −3) ∪ {−3} ∪ (−3, − 2) ∪ {− 2} ∪ (− 2, 2) ∪ { 2} ∪ ( 2, ∞) = R.
– The “sign-invariance” of P and Q over each region was exploited to allow only
a single sample point to be selected from each region. This property holds as
by the Intermediate Value Theorem, P and Q can only change sign by passing
through a root.
– The signs of√univariate polynomials were evaluated at irrational real algebraic
points like 2 to determine the truth values of atomic formulas.
In creating our automatic proof procedure, all of this routine reasoning must,
of course, be formalised. Moreover, the isolation of polynomial roots (and thus
sign-invariant regions) and the sign determination for polynomials at real algebraic points are computationally expensive operations. Computer algebra systems
like Mathematica have decades of tuning in their implementations of these core
algebraic algorithms. To have a practical proof procedure, we wish to take advantage of these highly tuned external tools as much as possible. Let us next describe
how this can be done.

3 A Sketch of our Certificate-based Design
There is a rich history of certificate-based, sceptical integrations between proof
assistants and external solvers. Examples include John Harrison’s sums-of-squares
method [17] and the Sledgehammer [31] command in Isabelle.
Certificate-based approaches are motivated by many observations, including:
– External solvers are often highly tuned and run much faster than verified ones.
– Verification of certificates from external solvers is usually much easier than
finding them. Such verification ensures the soundness of the overall tactic.
– Switching between different external solvers does not require changes in formal
proofs.

Algorithm 1 Prove univariate universal formulas over reals
Require: F (x) is a quantifier-free formula over reals
Ensure: Return true if ∀x. F (x) holds
1: procedure universal(∀x. F (x))
2:
P ← extract polynomials from F (x)
. P ⊆ Z[X]
3:
roots ← real roots of P
. Roots returned by external programs
4:
samples ← construct sample points from roots
5:
if (∀x ∈ samples. F (x)) ∧ (roots are indeed all real roots of P) then
6:
return true
7:
end if
8: end procedure

Algorithm 1 sketches our idea for univariate universal formulas. In particular,
in line 3, we use external programs to return real roots of polynomials (i.e., P)
from the quantifier-free part of the formula (i.e., F (x)). Those roots (i.e., roots)
correspond to a decomposition such that each polynomial from P has a constant
sign over each component of this decomposition. Since the roots are returned by
untrusted programs, in line 5, we not only check ∀x ∈ samples. F (x) as in Equation
(1) but also certify that these roots are indeed all real roots of P.
The step in line 3 in Algorithm 1 is more commonly referred as (real) root
isolation, which is a classic and well-studied topic in symbolic computing. Although we can in principle formalise our own root isolation procedure (e.g., using
the Sturm-Tarski theorem), it is utterly unlikely that our implementation will be
competitive with state-of-the-art ones, especially for polynomials of high degree,
large bit-width, or whose roots are very close together. Therefore, we delegate this
computationally expensive step to external tools.
With existential formulas, the situation is even simpler as illustrated in Algorithm 2, since we do not need to deal with the decomposition internally. Rather, all
we need is a real algebraic witness that satisfies λx. F (x) to certify ∃x. F (x). What
is more interesting is that the satisfaction problem for λx. F (x) can be not only
solved by a CAD procedure, which is complete but not very fast due to its symbolic nature, but also be complemented by highly efficient incomplete numerical
methods. Thus it is natural to externalize the step in line 2 in Algorithm 2.

6

Wenda Li et al.

Algorithm 2 Prove univariate existential formulas over reals
Require: F (x) is a quantifier-free formula over reals
Ensure: Return true if ∃x. F (x) holds
1: procedure existential(∃x. F (x))
2:
r ← solution to F (x)
. Solution returned by external programs
3:
if F (r) then
4:
return true
5:
end if
6: end procedure

4 Encoding Real Algebraic Numbers
External
√ programs in either Algorithm 1 and 2 can return real algebraic numbers
(e.g. 2). In this section, we see how to formalise such numbers in Isabelle/HOL.
The real algebraic numbers (Ralg ) are real roots of non-zero polynomials with
integer (equivalently, rational) coefficients. They form a countable, computable
subfield of the real numbers. To encode them, we use a polynomial with integer coefficients and a root selection method to “pin down” the root in question.
Common root selection methods include isolating intervals, root indices or Thom
encodings. We use the root interval approach, that is, a real algebraic number
r ∈ Ralg will be given by
– a polynomial p ∈ Z[x] s.t. p(r) = 0, and
– two rationals a, b ∈ Q s.t. r is the only root of p contained in [a, b].
To reason over the reals, we define a function Alg to embed those real algebraic
numbers into the reals:
Alg:: "int poly ⇒ float ⇒ float ⇒ real"

where int poly is a polynomial with integer coefficients and the two float arguments represent an interval. Note, a float in Isabelle/HOL is a dyadic rational
number of the form
a2b where a, b ∈ Z.
Compared to our previous work [21], where a pair of rational numbers is used
to represent an interval, the dyadic rational approach is more efficient due to the
elimination of ubiquitous greatest common divisor (gcd) operations within rational
arithmetic.
In Isabelle/HOL, a real number is represented as a Cauchy sequence of type
nat ⇒ rat, where a Cauchy sequence is defined as
definition
cauchy :: "(nat ⇒ rat) ⇒ bool"
where
"cauchy X ←→ ( ∀ r>0. ∃ k. ∀ m ≥k. ∀ n ≥k. |X m - X n | < r)"

We then convert an encoding of a real algebraic number into a sequence of type
nat ⇒ rat. The idea is to bisect the isolating interval through each recursive call,
and proceed with the half where the sign of the polynomial changes at its end
points:
fun to_cauchy:: "rat poly × rat × rat ⇒ nat ⇒ rat" where
"to_cauchy (_, lb, ub) 0 = (lb+ub)/2"|
"to_cauchy (p, lb, ub) (Suc n) = (

let c=(lb+ub)/2
in if poly p lb * poly p c ≤ 0
then to_cauchy (p, lb, c) n
else to_cauchy (p, c, ub) n)"

where poly p x evaluates the polynomial p at the point x. Note, rat poly × rat ×
rat encodes a real algebraic number here (rather than int poly × float × float ),
as we can embed int and float into rat.
It can be then shown that the sequence constructed by to cauchy (p, lb, ub)
is indeed a Cauchy sequence and the real number represented by this sequence
resides within the interval [lb, ub], provided lb < ub:
lemma to_cauchy_cauchy:
fixes p::"rat poly" and lb ub ::rat
assumes "lb<ub"
defines "X ≡to_cauchy (p,lb,ub)"
shows "cauchy X"
lemma to_cauchy_bound:
fixes p::"rat poly" and lb ub ::rat
defines "X ≡to_cauchy (p,lb,ub)"
assumes "lb<ub"
shows "lb ≤ Real X" "Real X ≤ ub"

Note, the function Real of type (nat ⇒ rat) ⇒ real constructs a real number from
its underlying representation (i.e. a Cauchy sequence).
Finally, we can finish the definition of Alg :
definition valid_alg::"int poly ⇒ float ⇒ float ⇒ bool" where
"valid_alg p lb ub = (lb < ub ∧ poly p lb * poly p ub < 0
∧ card ({x::real. poly p x = 0 ∧ lb < x ∧ x < ub}) = 1)"
definition Alg:: "int poly ⇒ float ⇒ float ⇒ real" where
"Alg p lb ub = (if valid_alg p lb ub
then Real (to_cauchy (p, lb, ub))
else undefined)"

where valid alg p lb ub ensures
– lb < ub,
– the polynomial p is of different signs (and non-zero) at lb and ub,
– the polynomial p has exactly one real root within the interval (lb,ub).
√
With the help of Alg, we can now encode the real algebraic number 2 as
Alg [:-2,0,1:] 1 2

where [:-2,0,1:] corresponds to the polynomial −2x0 + 0x1 + 1x2 = x2√− 2, and
1 and 2 are the lower bound and upper bound respectively, such that 2 is the
only root of x2 − 2 within the interval (1, 2).
Furthermore, we can formally derive that Alg p lb ub is indeed a root of p
within the interval (lb,ub) :
lemma alg_bound_and_root:
fixes p::"int poly" and lb ub::float
assumes "valid_alg p lb ub"
shows "lb < Alg p lb ub" and "Alg p lb ub < ub"
and "poly (of_int_poly p) (Alg p lb ub) = 0"

where of int poly p embeds the integer polynomial p into a real one.

5 Deciding the Sign of a Univariate Polynomial at Real Algebraic
Points
In the previous section, we described how to encode a real algebraic number as an
integer polynomial and two dyadic rational numbers. Now, suppose we have
√
2 = (x2 − 2, 1, 2)
where (x2 −2, 1, 2) is abbreviated from Alg [:-2,0,1:] 1 2 for the sake of readability.
How can we computationally prove that
√
P ( 2) = 0

where

P (x) =

1 2
x −1 ?
2

Considering that Ralg is a computable subfield of R and has decidable arithmetic
and comparison operations, it is natural to evaluate such formulas through algebraic arithmetic:
√
P ( 2)
1
= ×alg (x2 − 2, 1, 2) ×alg (x2 − 2, 1, 2) −alg 1
2
1
= ×alg (x − 2, 1, 3) −alg 1
2
1 3
= (x − 1, , ) −alg 1
2 2
= 0,
where ×alg and −alg are exact algebraic arithmetic operations that usually involve
calculation of bivariate resultants. Although such computations are currently possible in Isabelle/HOL [36, 21], they are far from efficient.
In this section, we describe a verified procedure to decide the sign of univariate polynomials with rational coefficients at real algebraic points which uses only
rational (or dyadic rational) arithmetic rather than costly algebraic arithmetic.

5.1 The Sturm-Tarski Theorem
We abbreviate R ∪ {−∞, ∞} as R, the extended real numbers.
Definition 1 (Tarski Query) The Tarski query TaQ(Q, P, a, b) is
TaQ(Q, P, a, b) =

X

sgn(Q(x))

x∈(a,b),P (x)=0

where a, b ∈ R, P, Q ∈ R[X], P 6= 0 and sgn : R → {−1, 0, 1} is the sign function.
The Sturm-Tarski theorem [23, Chapter 8] (or Tarski’s theorem [2, Chapter 2])
is essentially an effective way to compute Tarski queries through some remainder
sequences:

Theorem 1 (Sturm-Tarski) The Sturm-Tarski theorem states
TaQ(Q, P, a, b) = Var(SRemS(P, P 0 Q); a, b)
where P 6= 0, P, Q ∈ R[X], P 0 is the first derivative of P , a, b ∈ R, a < b and are
not roots of P , SRemS(P, P 0 Q) is the signed remainder sequence of P and P 0 Q,
and
Var([p0 , p1 , ..., pn ]; a, b)
= Var([p0 (a), p1 (a), ..., pn (a)]) − Var([p0 (b), p1 (b), ..., pn (b)])
is the difference in the number of sign variations (after removing zeroes) in the
polynomial sequence [p0 , p1 , ..., pn ] evaluated at a and b.
Note that the more famous Sturm’s theorem, which counts the number of
distinct real roots (of a univariate polynomial) within an interval, is a special case
of the Sturm-Tarski theorem when Q = 1.

5.2 A Formal Proof of the Sturm-Tarski Theorem
Our proof of the Sturm-Tarski theorem in Isabelle is based on Basu et al. [2,
Chapter 2] and Cohen’s formalisation in Coq [6].
The core idea of our formal proof is built around the Cauchy index. First
defined by Cauchy in 1837, the Cauchy index of a real rational function encodes
deep properties of its roots and poles, and can be used as the basis of an algebraic
method for computing Tarski queries3 .
Definition 2 Given P, Q ∈ R[x] and x ∈ R, jump(P, Q, x) is defined as

Q(u)
Q(u)

−1 if limu→x− P (u) = ∞ and limu→x+ P (u) = −∞
Q(u)
jump(P, Q, x) = 1
if limu→x− P (u) = −∞ and limu→x+ Q(u)
P (u) = ∞


0
otherwise.
For example, let Q = x − 4 and P = (x − 3)(x − 1)2 (x + 1). The graph of Q/P is
shown in Fig. 2. We have


when x = −1
1
jump(P, Q, x) = −1 when x = 3


0
otherwise.
The Cauchy index cindex poly a b q p is the sum of the jumps of q/p over the
interval (a, b):
definition cindex_poly:: "real ⇒ real ⇒ real poly ⇒ real poly ⇒ int"
where
P
"cindex_poly a b q p ≡ ( x ∈{x. poly p x=0 ∧ a < x ∧ x < b}. jump_poly q p x)"
3 Besides the application described in this section, the Cauchy index also plays a critical
role in the Routh–Hurwitz theorem. Interested readers may consult [32, Chapter 10,11] for
historical notes.

Fig. 2: Graph of the rational function (x − 4)/((x − 3)(x − 1)2 (x + 1))

By case analysis, we can prove a connection between the Tarski query and the
Cauchy index:
lemma cindex_poly_taq:
fixes p q::"real poly" and a b::real
shows "taq {x. poly p x = 0 ∧ a < x ∧ x < b} q
=cindex_poly a b (pderiv p * q) p"

where taq is a formal definition of the Tarski query
definition taq P
:: "’a::linordered_idom set ⇒ ’a poly ⇒ int" where
"taq s q = ( x ∈s. sign (poly q x))"

and pderiv p is the first derivative of p.
Moreover, the Cauchy index can be related to Euclidean division (mod ) on
polynomials by a recurrence:
cindex_poly_rec:
fixes p q::"real poly" and a b::real
assumes "a < b" and "poly (p * q) a 6=0"
and "poly (p * q) b 6=0"
shows "cindex_poly a b q p = cross (p * q) a b
+ cindex_poly a b (- (p mod q)) q"

where


0
cross p a b = 1


−1

if p(a)p(b) ≥ 0
if p(a)p(b) < 0 and p(a) < p(b)
if p(a)p(b) < 0 and p(a) ≥ p(b).

A similar recurrence relation holds for the number of sign variations of the signed
remainder sequences (changes itv smods ):

Univariate Decision Procedure

11

lemma changes_itv_smods_rec:
fixes p q::"real poly" and a b::real
assumes "a < b" and "poly (p * q) a 6= 0"
and "poly (p * q) b 6= 0"
shows "changes_itv_smods a b p q = cross (p * q) a b
+ changes_itv_smods a b q (- (p mod q))"

where changes itv smods is defined as
definition changes_itv_smods::
"real ⇒ real ⇒ real poly ⇒ real poly ⇒ int" where
"changes_itv_smods a b p q = (
let
ps = smods p q
in
changes_poly_at ps a - changes_poly_at ps b)"

and the signed remainder sequence (smods ) is defined as
function smods:: "real poly ⇒ real poly ⇒ (real poly) list" where
"smods p q= (if p = 0 then
[]
else
p # (smods q (-(p mod q))))"

and changes poly at ps a returns the number of sign changes when evaluating a list
of polynomials (ps ) at a.
Finally, by combining cindex poly taq, cindex poly rec and changes itv smods rec,
we derive the Sturm-Tarski theorem:
theorem sturm_tarski_interval:
fixes p q::"real poly" and a b::real
assumes "a < b" and "poly p a 6= 0" and "poly p b 6= 0"
shows "taq {x. poly p x = 0 ∧ a < x ∧ x < b} q
= changes_itv_smods a b p (pderiv p * q)"

Note, this is just the bounded case of the Sturm-Tarski theorem. Proofs for
the unbounded and half-bounded cases are similar.

5.3 Sign Determination through the Sturm-Tarski Theorem
Given a polynomial q with rational coefficients and our encoding of a real algebraic
number α
α = (p, lb, ub)
where p is an integer polynomial, and lb and ub are dyadic rationals, we can effectively decide the sign of q(α) using the Sturm-Tarski theorem, provided valid alg
p lb ub holds. The rationale behind is that valid alg p lb ub ensures α is the only
root of p within the interval (lb, ub), hence
X
sgn(q(α)) =
sgn(q(x))
x∈(lb,ub),p(x)=0

= TaQ(q, p, lb, ub)
= Var(SRemS(p, p0 q); lb, ub).
Importantly, it can be observed that evaluating Var(SRemS(p, p0 q); lb, ub) requires
only rational arithmetic rather than costly algebraic arithmetic.

To be even more efficient, we refine the procedure further to make use of dyadic
rational arithmetic. The main advantage of dyadic rational arithmetic over rational
arithmetic are reduced normalization steps and possible bit-level operations. For
example, consider two rational numbers ab11 and ab21 where a1 , b1 , a2 , b2 ∈ Z, their
sum is
a1
a2
a1 b2 + a2 b1
(a1 b2 + a2 b1 )/c
+
=
=
b1
b2
b1 b2
(b1 b2 )/c
where c = gcd(a1 b2 + a2 b1 , b1 b2 ).
To counter the growth in the size of representations, we usually need to normalize
the result by factoring out the gcd. Such gcd operations can be the source of
major computational expense. Thankfully, they are unnecessary in the context
of dyadic rationals. The sum of two dyadic rationals (a1 , e1 ) and (a2 , e2 ) where
a1 , e1 , a2 , b2 ∈ Z is
(
(a1 2e1 −e2 + a2 )2e2 if e1 > e2
e1
e2
a1 2 + a2 2 =
(a1 + a2 2e2 −e1 )2e1 otherwise.
Moreover, multiplications by powers of two, such as a1 2e1 −e2 , can be optimised
by shift operations.
However, the problem with dyadic rational numbers is that they do not have
the division operation (e.g. 1 × 20 divided by 3 × 20 is no longer a dyadic rational),
hence they do not form a field, while Euclidean division only works for polynomials
over a field. This problem can be solved if we switch from Euclidean division (mod
and div ):
P = (P div Q) Q + (P mod Q) and (Q = 0 ∨ deg(P mod Q) < deg(Q))
to pseudo-division (pmod and pdiv ) [10]:
lc(Q)1+deg(P )−deg(Q) P = (P pdiv Q) Q + (P pmod Q)
and (Q = 0 ∨ deg(P mod Q) < deg(Q))
where lc(Q) is the leading coefficient of Q,
since pseudo-division can be carried out by polynomials over an integral domain
(rather than a field).
Based on pseudo-division, the signed pseudo-remainder sequence (SPRemS)
can be defined:
function spmods :: "’a::idom poly ⇒ ’a poly ⇒ (’a poly) list" where
"spmods p q= (if p=0 then [] else
let
m=(if even(degree p+1-degree q) then -1 else -lead_coeff q)
in
Cons p (spmods q (smult m (p pmod q))))"

where smult is the scalar product on polynomials and lead coeff q is the leading
coefficient of q. Accordingly, the function to count the difference in sign variations
can be refined:

definition changes_itv_spmods::
"’a ::linordered_idom ⇒ ’a ⇒ ’a poly ⇒ ’a poly ⇒ int" where
"changes_itv_spmods a b p q= (let ps = spmods p q in
changes_poly_at ps a - changes_poly_at ps b)"

and linked to the previous one based on signed remainder sequences (SRemS):
lemma changes_spmods_smods:
fixes p q::"float poly" and a b::"float"
shows "changes_itv_spmods a b p q
= changes_itv_smods (real_of_float a) (real_of_float b)
(of_float_poly p) (of_float_poly q)"

where real of float embeds a float into real and of float poly coverts a float poly
(i.e. polynomial with dyadic rational coefficients) to a real poly by embedding
each of the coefficients into real.
Finally, we define a function sgn at that returns the sign of a univariate polynomial at some point:
definition "(sgn_at::real poly ⇒real ⇒real) = ( λq x. sgn (poly q x))"

Note, for now, if either x or any coefficient of q is an irrational real number (e.g.
an irrational real algebraic number), evaluating sgn at q x will raise an exception,
as Isabelle/HOL, by default, only supports rational arithmetic. Although we can
eliminate some such exceptions by loading any of the recent algebraic arithmetic
libraries [36, 21], we consider exact algebraic arithmetic too slow for our purpose as
stated at the beginning of Sec. 5. Alternatively, by proving some code equations,
we can restore the executability of sgn at q x when x is constructed by Alg p lb ub
and coefficients of q are rational reals:
lemma sgn_at_code_alg[code]:
fixes q::"real poly" and p::"int poly" and lb ub::float
shows "sgn_at q (Alg p lb ub) = (
if valid_alg p lb ub ∧ ( ∀ x ∈set (coeffs q). is_rat x) then
(let
p’::float poly=of_int_poly p;
q’::float poly=of_int_poly (int_poly q)
in
of_int (changes_itv_spmods lb ub p’ (pderiv p’ * q’)))
else Code.abort (STR ’’Invalid sgn_at’’)
( λ_. sgn_at q (Alg p lb ub)))"

where
– ∀ x ∈set (coeffs q). is_rat x checks if each coefficient of q is rational,
– of int poly converts an integer polynomial into a dyadic rational one,
– int poly clears denominators in the coefficients by multiplying each coefficient
by the least common multiple (of the denominators),
– Code.abort throws an exception, if either (p, lb, ub) is an invalid representation
of a real algebraic number or the polynomial q has any non-rational coefficient.
And note that evaluating changes_itv_spmods lb ub p’ (pderiv p’ * q’) requires only
dyadic arithmetic, which is much more efficient than exact algebraic arithmetic.
Moreover, the executability of valid alg is restored similarly as well:
lemma [code]:
fixes p::"int poly" and lb ub::float
shows "valid_alg p lb ub = (lb < ub
∧ (sgn (poly (of_int_poly p) lb) * sgn (poly (of_int_poly p) ub) < 0)
∧ changes_itv_spmods lb ub (of_int_poly p) (pderiv (of_int_poly p)) = 1)"

where
changes_itv_spmods lb ub (of_int_poly p) (pderiv (of_int_poly p)) = 1

checks if the polynomial p has exactly one real root within the interval (lb, ub) by
exploiting Sturm’s theorem (a special case of our formalised Sturm-Tarski theorem).
After restoring executability of sgn
√ at on real algebraic numbers, we can now
check the sign of P (x) = 12 x2 − 1 at 2 by typing the following command:
value "sgn_at [:-1,0,1/2:] (Alg [:-2,0,1:] 1 2)"

√
which returns 0 (i.e. P ( 2) = 0).

5.4 Remark
A formal proof of the Sturm-Tarski theorem is not new among proof assistants: it
has been formalised in PVS [25] and Coq [6]. However, as far as we know, we are
the first to exploit this theorem to build a verified sign determination procedure
of real algebraic numbers, which uses only rational or dyadic rational arithmetic.
Real algebraic numbers are essential in symbolic computing, and well studied.
In general, exact real algebraic arithmetic is rarely used in modern computer algebra systems due to its extreme inefficiency. For example, consider the problem
of isolating the real roots of a polynomial with real algebraic coefficients. Modern
approaches usually use sophisticated techniques to soundly approximate those coefficients to a certain precision rather than carrying out exact algebraic arithmetic
[33, 5, 35], relying on exact symbolic procedures as a fall-back in degenerate cases.
Following these efficient modern approaches, our sign determination procedure
can be improved in at least the following ways:
– Sophisticated interval arithmetic can be used to decide the sign before resorting
to a remainder sequence, as has been done in Z3 [10]. This approach should
help when the sign is non-zero.
– Pseudo-division, which we are currently using for building remainder sequences,
is not good for controlling coefficients growth. More sophisticated approaches,
such as subresultant sequences and modular methods, can be used to optimise
the calculation of remainder sequences.

6 The Formal Development of the Decision Procedure
In this section, we describe the main proof underlying our tactic.

6.1 Parsing Formulas
The first step of our tactic is to parse the target formula into a structured form.
This process is usually referred as reification [4] in Isabelle/HOL. More specifically,
given an Isabelle/HOL term e of type τ , we define a (more structured) datatype δ
and an interpretation function interp of type δ ⇒ τ list ⇒ τ , such that for some
e‘ of type δ
e = interp e‘ xs

where xs is a list of free variables in e. Subsequently, instead of directly dealing
with e, we now convert it into a more pleasant form interp e‘ xs where e‘ is in fact
a formal language that captures the structure of e.
The datatypes we defined to capture the structure of target univariate formulas
are as follows:
datatype num = C real — Constant
| Var nat — Variable index
| Add num num | Minus num | Mul num num | Power num nat
datatype norm_num2 =
Pol "int poly" nat — an integer polynomial and its variable index
| Const real — constant
| Abnorm num — in case of anomalies (e.g., bivariate)
datatype qf_form2 =
Pos norm_num2 — is positive | Zero norm_num2 — is zero
| Neg qf_form2 — negation
| Conj qf_form2 qf_form2 — conjunction
| Disj qf_form2 qf_form2 — disjunction
| T — true | F — false
datatype norm_form2 =
QF qf_form2 — quantifier free
| ExQ norm_form2 — existential
| AllQ norm_form2 — universal

and the interpretation functions:
fun num_interp:: "num ⇒ real list ⇒ real" where
"num_interp (C i) vs = i"|
"num_interp (Var v) vs = vs!v"|
"num_interp (Add num1 num2) vs = num_interp num1 vs + num_interp num2 vs "|
"num_interp (Minus num) vs = - num_interp num vs "|
"num_interp (Mul num1 num2) vs = num_interp num1 vs * num_interp num2 vs "|
"num_interp (Power num n) vs = (num_interp num vs)^n"
fun norm_num2_interp :: "norm_num2 ⇒ real list ⇒ real" where
"norm_num2_interp (Pol p v) vs = poly (of_int_poly p) (vs!v)"|
"norm_num2_interp (Const c) vs = c"|
"norm_num2_interp (Abnorm num) vs = num_interp num vs" — anomaly
fun qf_form2_interp:: "qf_form2 ⇒ real list ⇒ bool" where
"qf_form2_interp (Pos norm_num) vs = (norm_num2_interp norm_num vs > 0)"|
"qf_form2_interp (Zero norm_num) vs = (norm_num2_interp norm_num vs = 0)"|
"qf_form2_interp (Neg qf_form) vs = ( ¬ qf_form2_interp qf_form vs)" |
"qf_form2_interp (Conj qf_form1 norm_form2) vs
= (qf_form2_interp qf_form1 vs ∧ qf_form2_interp norm_form2 vs)"|
"qf_form2_interp (Disj qf_form1 qf_form2) vs
= (qf_form2_interp qf_form1 vs ∨ qf_form2_interp qf_form2 vs)"|
"qf_form2_interp T vs = True"|
"qf_form2_interp F vs = False"
fun norm_form2_interp:: "norm_form2 ⇒real list ⇒ bool" where
"norm_form2_interp (QF qf) vs = qf_form2_interp qf vs"|
"norm_form2_interp (ExQ norm_form) vs
= ( ∃ x. norm_form2_interp norm_form (x#vs))"|
"norm_form2_interp (AllQ norm_form) vs
= ( ∀ x. norm_form2_interp norm_form (x#vs))"

Given the definition of a (structured) datatype norm form2 and the corresponding interpretation function norm form2 interp, target formulas can now be parsed.
For example, we can convert a univariate formula

" ∀ x::real. x > 1/2 ∨ x < 1"

into an equivalent form
"norm_form2_interp
(AllQ (QF (Disj (Pos (Pol [:- 1, 2:] 0))
(Pos (Pol [:1, - 1:] 0))
)))
[]"

In particular, note
qf_form2_interp (Pos (Pol [:- 1, 2:] 0)) [x]
= (poly [:- 1, 2:] x > 0)
= (x > 1/2)

in which inequalities have been parsed into a polynomial sign determination problem.
On the contrary, a bivariate non-closed formula such as
" ∃ x::real. x + y >0"

will be converted into
"norm_form2_interp
(ExQ (QF (Pos
(Abnorm
(Add (Add (Add (C 0) (Mul (Var 0) (Add (C 1) (Mul (Var 0) (C 0)))))
(Add (C 0) (Mul (Var 1) (Add (C 1) (Mul (Var 1) (C 0))))))
(C 0))))))
[y]"

where the Abnorm constructor indicates that such formula is not supported by our
current tactic.
6.2 Existential Case
To discharge a univariate existential formula is easy: we can computationally check
if a certificate (i.e., a real algebraic number) returned by an external solver satisfies
the quantifier-free part of the formula:
lemma ExQ_intro:
fixes x::"alg_float" and qf_form::qf_form2
assumes "qf_form2_interp qf_form [of_alg_float x]"
shows "norm_form2_interp (ExQ (QF qf_form)) []"

where x of type alg float
datatype alg_float =
Arep "int poly" float float — representation of a real algebraic number
| Flt float — a small optimization in case the number is dyadic rational

is a certificate that is supposed to be instantiated by an external solver. The
function of alg float converts x from alg float to real. In other words, to prove
an existential formula:
"norm_form2_interp (ExQ (QF qf_form)) []"

we can computationally check the truth value of the quantifier-free part of the
formula at x :
"qf_form2_interp qf_form [of_alg_float x]"

which is possible due to the sign determination procedure described in Sec. 5.

6.3 Universal Case
For the universal case, the core lemma is as follows:
lemma utilize_samples:
fixes P::"real ⇒ bool" and decomps::"real set set"
and samples::"real
set" and f::"real set ⇒ real"
S
assumes " decomps = IR"
and " ∀ d ∈decomps. ∀ x1 ∈d. ∀ x2 ∈d. P x1 = P x2"
and " ∀ d ∈decomps. f d ∈d" and "bij_betw f decomps samples"
shows "( ∀ x. P x) = ( ∀ pt ∈samples. P pt)"

where bij_betw f decomps samples states that f::real set ⇒ real is a bijective function between the decomposition decomps::real set set and the sample points samples::real
set. Essentially, what the lemma utilize samples shows is that given a predicate P::real ⇒ bool, an unbounded universal formula ∀ x. P x is equivalent to
a bounded one ∀ pt ∈samples. P pt, if the truth value of P is constant over each
component of the decomposition: ∀ d ∈decomps. ∀ x1 ∈d. ∀ x2 ∈d. P x1 = P x2.
On top of the lemma utilize samples, we similarly convert an unbounded univariate real formula into a bounded one:
lemma allQ_subst:
fixes root_reps::"alg_float list" and pols::"float poly set"
and qf_form::qf_form2
defines "samples ≡map of_alg_float (mk_samples root_reps)"
assumes "Some pols = extractPols qf_form"
and "ordered_reps root_reps"
and "contain_all_roots root_reps pols"
and "valid_list root_reps"
shows "norm_form2_interp (AllQ (QF qf_form)) vs
= ( ∀ x ∈ (set samples). norm_form2_interp (QF qf_form) (x#vs))"

where
– root reps::alg float list is a certificate that should be instantiated by an external solver. More specifically, root reps should be the representation of a list
of real roots (in ascending order) of polynomials from the quantifier-free part
of the target formula,
– map of_alg_float (mk_samples root_reps) constructs sample points from the representation of a list of roots,
– extractPols qf form extracts polynomials from the quantifier-free part qf form,
– ordered reps root reps and valid list root reps together ensure that the representation of roots are valid and those roots are in ascending order,
– contain all roots roots reps pols checks if root reps is a representation of all real
roots of the polynomials pols. Specifically, by Sturm’s theorem, the number of
total distinct real roots of each p ∈ pols can be computed, which can be then
compared with the number of r ∈ root reps that p(r)=0.
Most importantly, all assumptions of the lemma allQ subst and its right-hand side
( ∀ x ∈ (set samples). norm_form2_interp (QF qf_form) (x#vs))

can be computationally checked, through which we can prove an unbounded univariate universal formula: norm_form2_interp (AllQ (QF qf_form)) vs.

7 Linking to an External Solver
Certificates for both existential and universal cases can be produced by any program performing univariate CAD. For now, we implement the program on top
of Mathematica. More specifically, the universal certificates are constructed by
the Mathematica command SemialgebraicComponentInstances, which gives sample points in each connected component of a semialgebraic set. The existential
certificates are constructed by the command FindInstance, which incorporates
powerful numerical methods to accelerate the search for real algebraic sample
points.
Also, it may be worth mentioning that after a certificate has been found, our
tactic will record it (as a string) so that repeating the proof no longer requires the
external solver. This is much like the sums-of-squares tactic [17].
In general, the certificate-based design grants us much flexibility: We can easily
switch to a more efficient external solver without modifying existing formal proofs.
In fact, we were first using an implementation of univariate CAD built within
MetiTarski, which turned out to be not very efficient, and we simply switched to
the current one based on Mathematica. In the future, we plan to experiment with
other open-source CAD implementations such as Z3 and QEPCAD to provide
more options with external solvers.

8 Experiments and Related Work
The most relevant work is the recent tarski strategy by Narkawicz et al. [25]
in PVS. Both their work and ours rely on a formal proof of the Sturm-Tarski
theorem (which they call Tarski’s theorem) and handle roughly the same class
of problems4 (i.e., first-order univariate formulas over reals). There are two main
differences between their work and ours:
– Their procedure resembles Tarski’s original quantifier elimination [2, Chapter 2]
and Cyril Cohen’s quantifier elimination procedure in Coq [6, Chapter 12] by
making use of both the Sturm-Tarski theorem and matrices. In contrast, our
tactic is based on CAD and real algebraic numbers (instead of matrices).
– Their procedure is entirely built within PVS, while ours sceptically makes use
of efficient external programs to generate certificates.
To compare both tactics empirically, we have conducted experiments on several
typical examples from their paper5 and the MetiTarski project6 [29]. The experiments are run on a desktop with an Intel Core 2 Quad Q9400 (quad core, 2.66
GHz) CPU and 8 gigabytes RAM. Results of the experiments are illustrated in
Fig. 3, where our univ rcf tactic includes both certificate searching and checking
process, while the univ rcf cert does the checking part only (when repeating a
proof with certificates already recorded as a string).
In general, the experiments indicate that our tactic outperforms the tarski
strategy in PVS. Particularly, the advantage of our tactic becomes greater as the
4 In fact, their tactic does not handle arbitrary boolean expressions like ours, but we believe
this should not be too hard to overcome.
5 http://shemesh.larc.nasa.gov/people/cam/Tarski/
6 http://www.cl.cam.ac.uk/ gp351/cicm2012/
~

Formula

univ rcf cert (Isabelle)

ex1
0.9
0.3
2.0
ex2
1.4
0.6
6.8
ex3
1.6
0.7
13.0
ex4
1.3
0.5
20.1
ex5
1.6
0.6
315.7
ex6
5.6
3.9
timeout
ex7
38.4
34.9
timeout
Note: timeout indicates failure to terminate within 24 hours

Fig. 3: Comparison between our tactic in Isabelle and the tarski strategy in PVS:
univ rcf includes certificate searching and checking, while univ rcf cert includes
only checking

problems become more complex, which can be attributed to the fact that our tactic
has much better worst-case computational complexity (polynomial vs. exponential
in the number of polynomials).
In the case of general multivariate problems, the CAD procedure is doubly
exponential while Tarski’s quantifier elimination procedure is non-elementary in
the number of variables [2, Chapter 11]). When limited to univariate problems,
the CAD procedure degenerates to root isolation and sign determination on a
set of univariate polynomials, which is of polynomial complexity in the number
of polynomials and their degree bound [2, Chapter 10]). In comparison, Tarski’s
quantifier elimination procedure, even when limited to univariate problems, is still
exponential in the number of polynomials [7].
In addition, it is worth noting that as the problems become more complex (e.g.,
ex6 and ex7 in Fig. 3), certificate checking becomes the bottleneck factor of our
tactic (especially for universal problems). This indicates that, despite the fact that

certificate searching is much harder than certificate checking, the Mathematica
implementation is still much more efficient than our verified certificate-checking
procedure. This leaves much room for future optimisations.
Our work has also been greatly inspired by Cyril Cohen’s PhD thesis [6], within
which a quantifier elimination procedure has been built upon the Sturm-Tarski
theorem and real algebraic numbers formalised within the Coq theorem prover.
However, our goals and approaches are very different.
Cohen’s work is part of a large project that has formalised the Feit-Thompson
theorem (odd order theorem) in Coq [15], and focuses more on theoretical developments than we do. For example, they proved the Sturm-Tarski theorem to
construct an RCF quantifier elimination procedure in the spirit of Tarski’s original
method, which has important theoretical properties but is not practical as a proof
procedure. Moreover, he has formalised arithmetic on real algebraic numbers and
shown that they form a real closed field via resultants. We have not formalised resultants at all. Our sign determination algorithm uses the Sturm-Tarski theorem,
which is significantly more efficient in practice than using resultants. On the other
hand, as it was unnecessary for our proof procedure, we have not proved in Isabelle
that the real algebraic numbers form a real closed field. In general, compared to
his work, ours stresses the practical side over the theoretical. Fundamentally, we
want to build procedures to solve non-trivial problems in practice.
Decision procedures based on Sturm’s theorem have been implemented in Isabelle and PVS before [14, 26]. Their core idea is to count the number of real roots
within a certain (bounded or unbounded) interval. Generally, they can only handle formulas involving a single polynomial, so they are not complete for first-order
formulas (unlike our tactic and the tarski strategy in PVS).
Assia Mahboubi [22] has implemented the executable part of a general CAD
procedure in Coq, but as far as we know, the correctness proof for her implementation is still ongoing. This is also one of the reasons for us to choose the
certificate-based approach rather than directly verifying an implementation.
There are other methods to handle nonlinear polynomial problems in theorem
provers, such as sums of squares [17], which is good for multivariate universal
problems but is not applicable when the existential quantifier arises, and interval
arithmetic [18, 34], which is very efficient for some cases but is not complete. These
methods and ours should be used in a complementary way.

9 Discussion and Applications
One of our driving motivations is the integration of MetiTarski with Isabelle.
MetiTarski [1] is a first-order theorem prover for real number inequalities involving
transcendental functions such as sin, tan and exp. It can automatically prove
formulas like
∀x ∈ (0, 1.25). tan(x)2 ≤ 1.75 × 10−7 + tan(1) tan(x2 )
1 − e−2x
1
1
− 2 ≤
2x(1 − e−x )2
x
12
√
√
x
1+x− 1−x
√
√
√
∀x ∈ (0, 1). 1.914
≤ 0.01 +
.
4+ 1+x+ 1−x
2 + 1 − x2
∀x > 0.

The main idea behind MetiTarski is to approximate transcendental functions
by polynomial or rational function bounds, and then solve the formula by a combination of a resolution theorem proving and an external Real Closed Field (RCF)
decision procedure (QEPCAD, Mathematica or Z3). MetiTarski is a version of
Joe Hurd’s Metis prover [19], modified to include arithmetic simplification and
integration with RCF decision procedures, along with many other refinements.
Applications of MetiTarski include verification problems arising in air traffic
control [13] and analogue circuit designs [11]. As some of the applications are safety
critical, it is natural to consider to integrate MetiTarski with an existing interactive
theorem prover, whose internal logic can be used to ensure the correctness of
MetiTarski’s proofs. Besides, the automation provided by MetiTarski is generally
useful to interactive theorem provers.
MetiTarski has been integrated with the PVS theorem prover [28] as a trusted
oracle [12]. The authors state that the automation introduced by MetiTarski for
closing sequents containing real-valued functions considerably outperforms existing tactics in PVS. However, this tactic should not be used in a certification
environment, where external oracles are not allowed.
Our eventual goal is to integrate MetiTarski into the Isabelle/HOL theorem
prover. Isabelle can verify purely logical inferences (in fact, it contains an internal
copy of the Metis theorem prover), and the third author has just formalised most
of the bounds of transcendental functions used by MetiTarski [30]. The primary
remaining hurdle is the RCF decision procedure, and the work presented here is
the first step towards it.
Finally, let us say a bit about how our work might be generalised to multivariate
problems. In doing so, we plan to continue our certificate-based approach, as we are
unlikely to implement a verified internal CAD procedure comparable in efficiency
to a state-of-the-art implementation. It is still not obvious to us where the clear
separation between search and verification should be in the multivariate case, but
we have already made some progress:
– The bivariate sign determination procedure based on recursive application of
the Sturm-Tarski theorem described in our previous work [21] can be easily
generalised to a multivariate one (i.e., a procedure to decide the sign of a
multivariate polynomial at real algebraic points), which can be then used to
efficiently certify purely existential multivariate formulas over reals.
– Our recent formalisation of Cauchy’s residue theorem [20] can be used to certify
a key theorem used in general CAD: that the complex roots of a polynomial
continuously depend on its coefficients.

10 Conclusion
We have described our work of building a procedure for first-order univariate
polynomial problems in Isabelle/HOL. Compared to existing tactics among proof
assistants, noticeable features of our tactic are
– It is based on univariate cylindrical algebraic decomposition (CAD).
– It sceptically integrates efficient external solvers in a certificate-based way,
so that its soundness solely depends on Isabelle’s logic (and code generation
machinery) rather than the external solvers.

This is made possible by certificate-based approaches to real root isolation and
sign-determination for evaluating polynomials at real algebraic points. As much
of the novelty in our work is motivated by practical efficiency considerations, we
have performed experiments comparing our procedure with another real algebraic
proof procedure, the tarski method in PVS. By making use of efficient external
solvers, our procedure is shown to empirically outperform this other method by
substantial margins. We believe this adds further impetus to the certificate-based
methods for a wide variety of formal proof procedures.
Certificate-based methods can be compared on the basis of how much mathematics and computation are required both to find and check their certificates.
For example, to convert a Positivstellensatz certificate into a HOL-Light proof of
a universal theorem, Harrison’s sums-of-squares tactic only requires simple signbased reasoning and rational arithmetic, while in our case, we need more mathematics (e.g., real algebraic numbers and the Sturm-Tarski theorem) and more
computation (especially for the universal case). A good certificate design needs to
balance the difficulty of the formalisation effort and verified computation required
to check the certificates with the efficiency improvements offered by offloading the
construction of the certificates to high-performance external tools.

